# BDDC v2

# A basic bdd-based logical calculator

# Pascal RAYMOND http://www-verimag.imag.fr/PEOPLE/Pascal.Raymond

#### November 24, 2008

BDDC is a tool for manipulating logical formula. It is based on a Binary Decision Diagram library, that means that each formula is internally tranformed into a canonical form. In particular, if the formula (resp its negation) is a tautologie, BDDC will transform it into the "always true formula" (resp. the "always false formula").

All classical operators are provided. Moreover the tool provides a "functional-like" language that allows the user to define his (her) own logical operators.

#### Contents

1	Des 1.1 1.2	cription         Basic features	2 2 2				
2	Lex	ical aspects	3				
3	Con	Commands					
4	ical expressions	4					
	4.1	Boolean operators	4				
	4.2	Control operators	5				
	4.3	Bdd related operators	5				
	4.4	Priorities	7				
5 Annexes		nexes	7				
	5.1	Notes on BDD size	7				
	5.2	Notes on identifiers	7				
	5.3	Notes on user-defined functions	9				
	5.4	Notes on Shannon decomposition	9				
	5.5	Notes on output format	9				

6	Examples				
	6.1	Programming the negation	10		
	6.2	Finding an implicant in a bdd	11		
	6.3	Programming the union	11		

# 1 Description

#### 1.1 Basic features

The bddc utility implements an interactive processor for the evaluation of logical formula. It reads commands from the standard input, and prints the result to the standard output. The main commands are those which involve Boolean expressions: basically, bddc computes a bdd (binary decision diagram [1, 2]), which is a canonical representation of the expression.

For instance, if the user types the expression " $\mathbf{x} + \mathbf{y} \cdot \mathbf{z}$ ;", the calculator interprets symbols  $\mathbf{x}$ ,  $\mathbf{y}$  and  $\mathbf{z}$  as Boolean arguments (i.e., basic propositions), and the whole expression as a Boolean function over those arguments. Note that "+" stands for logical or, and "." operator for logical and. The calculator builds the bdd corresponding to the expression, and then prints a polynomial representation of this canonical form; in the example, it simply outputs " $\mathbf{x} + \mathbf{y} \cdot \mathbf{z}$ " (useful, isn't it?).

If the expression is a tautology, its bdd is the always-true function: for instance, if the user types "x => (y => x);" (=> is the logical implication), the result is "true".

#### 1.2 Advanced features

The user may assign an expression to a variable, and then use this variable to build a new expression (">>" is the prompt symbol of the processor):

```
>>f := (y => x);
x + -y
>>x => f;
true
```

Moreover, the user may define new operators on bdds. For instance :

```
>>union(A, B) := A + B;
--new function: union, arity: 2
>>union(x.y, t);
x.y + t
```

Indeed, this example is quite stupid. Defining new operators becomes interesting when using control operators and recursion (see 5.3, 6).

# 2 Lexical aspects

The separator for <expr-list> and <ident-list> is the "," character.

Identifiers are any strings starting with a letter, and containing only letters, digits and underscore symbol, excepting the following keywords:

and	constrain	compare	cuts	cnf
del	dnf	else	exec	exist
false	forall	help	high	if
in	ite	implies	let	list
low	nor	not	or	print
quit	reed	restrict	root	set
size	syntax	then	true	xor

The sequence // starts a comment, which ends at the next new-line. The sequence /\* starts a comment, which ends at the next \*/ sequence.

#### 3 Commands

The processor is started by the command bddc, then the user enters an interactive session.

quit exits the processor.

**help** prints the list of commands.

exec "<command-file>" reads commands from a file.

**syntax** prints the syntax of logical expressions.

list prints the list of arguments and variables in the current context (see 5.2).

< logical-expr>; builds the bdd corresponding to the expression in the current context, and outputs a polynomial equivalent to this bdd.

<ident> := <logical-exp> ; builds the bdd corresponding to the expression in the current context, assigns it to the variable <ident>, and outputs a polynomial equivalent to this bdd.

 $< ident> (< ident \ list>) := < logical-exp>;$  defines a function, wich can be used latter in logical expression.

**print** < *ident*> print a polynomial equivalent to the bdd held by the variable < *ident*>. This command is obsolete: it is equivalent to type "< *ident*>;", since a identifier is a logical expression.

size < logical-expr>; prints the size of the bdd corresponding to < logical-expr>. This size is expressed as a number of "bdd nodes" (see 5.1).

[dnf, cnf, reed] < logical-expr>; those commands are quite equivalent to typing "< logical-expr>;", except that the output is printed (respectively) in disjunctive, conjunctive, or Reed-Muller normal form (see 5.5).

**set** [**dnf**, **cnf**, **reed**] sets the output mode (disjunctive, conjunctive, or Reed-Muller normal form see 5.5).

# 4 Logical expressions

The calculator provides all the classical Boolean operators, plus some other specially dedicated to the bdd implementation. The syntax is infixed, and the expression can be typed on more than one line, so don't forget the terminating symbol ";"!

#### 4.1 Boolean operators

false (or "0") always-false function.

**true** (or "1") always-true function.

<*ident*> reference to an argument or a variable (depending on the current context).

```
\mathbf{not} < expr > (\mathbf{or} "-< expr > ")  complement.
```

$$\langle expr \rangle$$
 or  $\langle expr \rangle$  (or " $\langle expr \rangle$ " +  $\langle expr \rangle$ ") disjunction.

$$\langle expr \rangle$$
 and  $\langle expr \rangle$  (or " $\langle expr \rangle$ ") conjunction.

$$\langle expr \rangle$$
 xor  $\langle expr \rangle$  (or " $\langle expr \rangle$ ") exclusive or.

 $\langle expr \rangle \Rightarrow \langle expr \rangle$  implication.

 $\langle expr \rangle = \langle expr \rangle$  equivalence.

ite(<expr>,<expr>,<expr>) functionnal "if then else"; "ite(f,g,h)" is equivalent to "f.g + -f.h".

forall < ident-list> < expr> universal quantifier.

exist < ident-list> < expr> existential quantifier.

# < expr-list> at most one true in the list.

nor < expr-list> no one true in the list.

**xor** < *expr-list*> exactly one true in the list.

#### 4.2 Control operators

Those operators allow the user to control the way expressions are evaluated.

 $\operatorname{compare}(<\operatorname{expr}>,<\operatorname{expr}>)$  returns 1 (the always-true function) if the two expressions are logically equivalent, 0 (the always-false function) otherwise.

implies(< expr>, < expr>) returns 1 if the first expression implies the second one, 0 otherwise.

 $\operatorname{cuts}(\langle expr \rangle, \langle expr \rangle)$  returns 1 if the intersection of the two expressions is not empty (i.e. their logical and is not 0), 0 otherwise.

if <*expr>* then <*expr>* else <*expr>* lazy "if": if the first operand is always-true (resp. always-false), the second (resp. third) operand is not evaluated; otherwise, it behaves as the ite operator.

let  $\langle ident \rangle := \langle exp \rangle$  in  $\langle exp \rangle$ ; local variable definition: the first expression is evaluated in a new context where  $\langle ident \rangle$  is undefined. Then, the second expression is evaluated in a new context where the first result is associated to  $\langle ident \rangle$ . The original context is then restaured, and the second result is returned (see 5.2 for details).

#### 4.3 Bdd related operators

The following operators are specific to the bdd representation.

restrict( $\langle expr \rangle$ ,  $\langle expr \rangle$ ) "restrict(f,g)" is a function equivalent to "f" if "g"; the result is simplier than "f" (less bdd nodes). The result is a function h satisfying:

$$if(f \wedge g = 0)$$
 then  $h = 0$ 

$$if(f \lor \neg g = 1) \ then \ h = 1$$

Otherwise, h is some function satisfying:

$$(f \land g) \Rightarrow h \Rightarrow (f \lor \neg g)$$

#### N.B. The result is undefined if the second operand is 0.

 $constrain(\langle expr \rangle, \langle expr \rangle)$  (also called the generalized cofactor) is a kind of "restrict" operation. The result is not garanted to be "smaller" than f, but it has an interesting (but not trivial) property: let h = constrain(f,g)

$$h(x) = f(x')$$
 where  $x' =$  "the closer point to x in g"

The precise definition of the "closer point ..." function is a little bit complicated (and depends on the variable ordering), but the main result is that the choice of x' does not depend on f.

N.B. The result is undefined if the second operand is 0.

 $\sup(\langle expr \rangle)$  the "or" of all argument which the expression depends on (support).

 $root(\langle expr \rangle)$  the "least" argument which the expression depends on.

root(<expr>, ..., <expr>) when called with more than one argument, the root function returns the "least" argument which one of the expressions depends on. This generalisation is quite useful when defining recusive functions on the bdd stucture.

$$high(\langle expr \rangle)$$

low(<expr>) (Bdd decomposition) every formula "f" satisfies "f = if root(f) then high(f) else low(f)", where neither high(f) nor low(f) depend on root(f) (see 5.4).

#### **Priorities** 4.4

Expressions can be enclosed into parenthesis. Some priorities are used to resolve ambiguities:

```
• not, exist, forall,
   • else
   • and
   • or
   • => (right associative)
   • =, xor (right associative)
   • in
For instance:
let z := t = v in - exist x if y then x else z + z = y \Rightarrow z;
```

```
is interpreted as:
let z := (t = v) in ((-(exist x (if y then x else z)) + z) = (y => z));
```

#### 5 Annexes

#### 5.1 Notes on BDD size

The tool uses a internal representation knowned as "signed binary decision diagrams" (SBDD). With this representation, a boolean function F and its negation  $\neg F$  are (almost) represented by the same internal structure (they differ only on a boolean flag). With classic BDDs, the internal representation of F and  $\neg F$  are disjoints. As a consequence, the cost of the negation is constant with SBDDs, since it is linear with classical BDDs. Moreover, it is known that this representation is always cheaper (in size and in time computation) than classical BDDs.

#### 5.2Notes on identifiers

The calculator deals with several kinds of identifiers:

arguments are the (global) free variables, i.e. identifiers that was once used without being first declared.

global variables are the identifiers introduced at the top level by a assignment command.

**local variables** are either formal parameters of a user-defined function, or identifiers introduced by a let operator.

The way an identifier is interpreted (i.e. the identifier's status) depends on the current context. Intuitively, at the top level, there is a global context, giving the status of arguments and global variables. The status of a global identifier is implicitly given by its first occurence: if it first appears in the left hand side of an assignment, the status is *variable*; if it first appears in an expression, the status is *argument*. One can type "list" to see the current global context:

```
>>x := a . b;
x := a . b
>>list
parameters are: a b
variables are: x
An argument cannot be re-used as a global variable; a variable, indeed, can
be re-assigned:
>>a := x;
can't assign "a", not a variable
>>x := a + b;
```

Arguments are **globals**, while the scope of variables can be restricted using the let ... in ... operator.

```
>>y := let z := c + d in z;
y := c + d
>>list --note that z has disappeared, but not c and d:
arguments are: a b c d
variables are: x y

If id is a variable, let id := ... shadows the current value of id:
>>y;
c + d
>>z := let y := (let y := e.f in y.g) in x + y;
z := a + b + e.f.g
>>y;
c + d
```

Moreover, a local declaration shadows the status of the identifier:

```
>>a;
a
>>let a := e.f in a;
e.f
>>a;
```

x := a + b

#### 5.3 Notes on user-defined functions

For the time being, no static verifications are made on user-defined functions. We can simply give the semantics for function calls: if the user has defined a function foo(i1, ..., ik) := exp;, then foo(e1, ..., en), where e1, ..., en are correct expressions, is a correct expression equivalent to:

```
let i1 := e1 in
...
let in := en in
    exp;
```

In particular, a user-defined function may use free identifiers, but it is indeed quite dangerous: depending on the context call, those identifiers will be interpretted either as arguments, global variables, or even local variables of the current scope!

#### 5.4 Notes on Shannon decomposition

The bdd decomposition is a special case of Shannon decomposition, where the choice of the decomposition variable is forced by the actual bdd representation. The general Shannon decomposition does not depend on the bdd structure, and can be obtained using logical operators; for all variable x and function f, we have  $f = (x \wedge f1) \vee (\neg x \wedge f0)$ , where (in bddc syntax):

```
f1 := exist x (x.f); -- or better: f1 := constrain(f,x);
f0 := exist x (-x.f); -- or better: f0 := constrain(f,-x);
```

#### 5.5 Notes on output format

By default, the tool prints the results in a disjunctive normal form (sum of "ands"). The idea is to make the result as "readable" as possible. The quality of the "result" dramaticaly depends on the internal order of the arguments. For example, depending on the order of its arguments, the expression "if A then B else C" can produce different (but equivalent) polynomials, such as "A.B + -A.C + B.C" or "B.A + -A.C". Indeed, the second form is "better" since it is more concise, but the problem of printing the simpliest polynomial equivalent to a given function is exponential. So, there is no guarantee on the "quality" of the print result.

The user can also prints the results in a conjonctive normal form (product of or's). This is the dual of the first form.

The Reed-Muller normal form consists in expressing the boolean functions as polynomials in "and" and "xor" (the always-true function "1" is also necessary). This form is canonical (modulo the commutativity and associativity of the operators). For instance:

```
>>reed (if A then B else C);
>>B.A <> A.C <> C
>>reed (not A);
>>A <> 1
```

This form is interesting (since it is canonical), but it can seem "strange" and unreadable for most people! Moreover, there is no "simple" relation between the size of the Reed-Muller form and a classical polynomial form: one form can be exponentially better or worse than the other:

```
>>A + B + C;
A + B + C
>>reed (A + B + C);
A.B.C <> A.B <> A.C <> A <> B.C <> B <> C
>>A <> B <> C;
A.B.C + A.-B.-C + -A.B.-C + -A.-B.C
>>reed (A <> B <> C);
A <> B <> C
```

# 6 Examples

#### 6.1 Programming the negation

The following function implements the Boolean negation. Indeed, using this function is quite ineffective, since the built-in not operator does the same thing in constant time, but this simple algorithm shows the general structure of any unary bdd-based operator.

```
negation(a) :=
  if compare(a, true) then false
  else if compare(a, false) then true
  else ite( root(a), negation(high(a)), negation(low(a)));
```

In order to understand this algorithm, remain that any bdd a is either:

- true, i.e. the always true formula,
- false, i.e. the always false formula,
- a ternary "expression" of the form ite(root(a), high(a), low(a)), where root(a) is an argument <sup>1</sup>, and high(a) and low(a) are two formulas not depending in root(a).

<sup>&</sup>lt;sup>1</sup>More precisely the identity formula associated to an argument

The algorithm first treat the basic cases, by **comparing** the parameter to the constants **true** and **false**, and after the genaral case, where the parameter still depends on some argument. In this case, the function is recursively called on the corresponding high and low parts, and the results are combined under a test of the root argument.

#### 6.2 Finding an implicant in a bdd

This is another example of unary function, that builds, if it exists, an monomial that implies **a**:

```
path(a) :=
  if compare(a, true) then true
  else if compare(a, false) then false
  else if compare(high(a), false) then (not root(a) and path(low(a)))
  else (root(a) and path(high(a)))
:
```

#### 6.3 Programming the union

The following function implements the Boolean or. The only interest is in showing how to build a binary operator. We first treat the basic cases, where either a, b are constants. Then, we treat the case where both a and b are "binary" bdds. In this last case, we use the binary extension of the root operator to get the "minimal" argument between root(a) and root(b), and call it min\_root. Then, the recursive calls are made by decomposing both parameters according to min\_root.

```
union(a, b) :=
  if (compare(a, true) or compare(b, false)) then a
  else if (compare(b, true) or compare(a, false)) then b
  else let min_root := root(a, b) in
    ite(min_root,
        union(constrain(a, min_root), constrain(b, min_root)),
        union(constrain(a, not min_root), constrain(b, not min_root))
    );
```

The use of the generalized cofactor in the recursive calls may seem a little bit tricky. It is justified by the fact that (for instance) either constrain(a, min\_root) = a if min\_root  $\neq$  root(a), or constrain(a, min\_root) = high(a) if min\_root = root(a).

#### References

[1] S. B. Akers. Binary decision diagrams. *IEEE Transactions on Computers*, C-27(6), 1978.

[2] R. E. Bryant. Graph-based algorithms for boolean function manipulation.  $IEEE\ Transactions\ on\ Computers,\ C-35(8):677-692,\ 1986.$